Angular momentum and linear momentum: circular cylindrical coordinates

Masatsugu Sei Suzuki
Department of Physics, SUNY at Binghamton
(Date: March 03, 2023)

1. Introduction

In the $|\mathbf{r}\rangle$ representation, we have the following relations,

$$\langle \mathbf{r}|\hat{\mathbf{p}}|\psi\rangle = \mathbf{p}\langle \mathbf{r}|\psi\rangle = \frac{\hbar}{i}\nabla\psi(\mathbf{r}),$$

for the linear momentum operator, and

$$\langle \mathbf{r}|\hat{\mathbf{L}}|\psi\rangle = \langle \mathbf{r}|\hat{\mathbf{r}}\times\hat{\mathbf{p}}|\psi\rangle = (\mathbf{r}\times\mathbf{p})\langle \mathbf{r}|\psi\rangle = \frac{\hbar}{i}(\mathbf{r}\times\nabla)\psi(\mathbf{r}).$$

The Schrodinger equation can be rewritten as

$$\langle \mathbf{r}|\hat{H}|\psi\rangle = \langle \mathbf{r}|\frac{1}{2\mu}\hat{\mathbf{p}}^2 + V(\hat{\mathbf{r}})|\psi\rangle = [-\frac{\hbar^2}{2\mu}\nabla^2 + V(\mathbf{r})]\langle \mathbf{r}|\psi\rangle = E\langle \mathbf{r}|\psi\rangle$$

So that the wave function and the differential operators in quantum mechanics can be expressed by differential operators in the $|\mathbf{r}\rangle$ representation. Is it possible to calculate equations including differential operators? The answer is yes. The Mathematica has an excellent function for that. One can make us of the pure function of Mathematica. We learned this technique from a famous textbook, Mathematica for Physics book, 2nd edition by R.L. Zimmerman and F.I. Olness (Addison-Wesley, 2002). Such differential operators can be realized in the Mathematica (combination of # and &). For example, the operator of the linear momentum ***p***, we use the following notations,

$$p_x := \frac{\hbar}{i}D[\#, x]\&, \qquad p_x[n] := Nest[p_x, \#, n]\&,$$

in the Mathematica. In this case, we have

$$p_x[\psi[x]] = \frac{\hbar}{i}\frac{\partial}{\partial x}\psi[x], \qquad p_x[n][\psi[x]] = \left(\frac{\hbar}{i}\right)^n \frac{\partial^n}{\partial x^n}\psi[x].$$

The syntax $D[\#, x]\&$ represents a pure function that will take a single argument (in our case, wave function $)\psi[x]$. The argument is inserted in slot number one represented by the symbol #. The symbol & is short hand notation for the Function function.

Using our original programs based on these techniques of Mathematica, , we will discuss the angular momentum and linear momentum, and so in the cylindrical co-ordinates.

2. Cylindrical coordinates



Fig. Cylindrical coordinates. The angle between the unit vectors $\mathbf{e}_\rho$ and $\mathbf{e}_z$ is ϕ.

The vector $\mathbf{r}$ is represented by

$$\mathbf{r} = \rho\mathbf{e}_\rho + z\mathbf{e}_z = \mathbf{e}_x\rho\cos\phi + \mathbf{e}_y\rho\sin\phi + \mathbf{e}_y z\ .$$

The unit vectors:

$$\mathbf{e}_\rho = \frac{\partial\mathbf{r}}{\partial\rho} = \mathbf{e}_x\cos\phi + \mathbf{e}_y\sin\phi$$

$$\mathbf{e}_\phi = \frac{1}{\rho}\frac{\partial \mathbf{r}}{\partial \phi} = -\mathbf{e}_x \sin\phi + \mathbf{e}_y \cos\phi$$

$$\mathbf{e}_z = \frac{\partial \mathbf{r}}{\partial z} = \mathbf{e}_z$$

which are dependent on ϕ. Using the matrix form, we get

$$\begin{pmatrix} \mathbf{e}_\rho \\ \mathbf{e}_\phi \\ \mathbf{e}_z \end{pmatrix} = \begin{pmatrix} \cos\phi & \sin\phi & 0 \\ -\sin\phi & \cos\phi & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} \mathbf{e}_x \\ \mathbf{e}_y \\ \mathbf{e}_z \end{pmatrix}$$

and

$$\begin{pmatrix} \mathbf{e}_x \\ \mathbf{e}_y \\ \mathbf{e}_z \end{pmatrix} = \begin{pmatrix} \cos\phi & -\sin\phi & 0 \\ \sin\phi & \cos\phi & 0 \\ 0 & 0 & 1 \end{pmatrix} \begin{pmatrix} \mathbf{e}_\rho \\ \mathbf{e}_\phi \\ \mathbf{e}_z \end{pmatrix}$$

Or

$$\mathbf{e}_x = \mathbf{e}_\rho \cos\phi - \mathbf{e}_\phi \sin\phi,$$

$$\mathbf{e}_y = \mathbf{e}_\rho \sin\phi + \mathbf{e}_\phi \cos\phi$$

$$\mathbf{e}_z = \mathbf{e}_z$$

We also have

$$\mathbf{e}_r = \frac{1}{r}\mathbf{r} = \frac{1}{r}(\rho \mathbf{e}_\rho + z\mathbf{e}_z)$$

$$d\mathbf{r} = \mathbf{e}_\rho d\rho + \mathbf{e}_\phi \rho d\phi + \mathbf{e}_z dz = \mathbf{e}_\rho h_\rho d\rho + \mathbf{e}_\phi h_\phi d\phi + \mathbf{e}_z h_z dz$$

with $h_\rho = 1,\ h_\phi = \rho,\ \ h_z = 1$

2. Differential operators; p and L

The orbital angular momentum $\mathbf{L}$ .s given by

$$\mathbf{L} = \mathbf{r} \times \mathbf{p} = \frac{\hbar}{i} \mathbf{r} \times \nabla ,$$

With the linear momentum as

$$\mathbf{p} = \frac{\hbar}{i} \nabla = \frac{\hbar}{i} \mathrm{Grad} .$$

We note that

$$L_x = \mathbf{e}_x \cdot \mathbf{L} , \qquad L_y = \mathbf{e}_y \cdot \mathbf{L} \qquad L_z = \mathbf{e}_z \cdot \mathbf{L}$$

The raising operator:

$$L_+ = L_x + iL_y$$

The lowering operatot:

$$L_- = L_x - iL_y$$

The radial linear momentum is defined by

$$p_r = \frac{1}{2} (\frac{\mathbf{r}}{r} \cdot \mathbf{p} + \mathbf{p} \cdot \frac{\mathbf{r}}{r}) = \frac{\hbar}{2i} (\frac{\mathbf{r}}{r} \cdot \nabla + \nabla \cdot \frac{\mathbf{r}}{r})$$

Note that

$$p_x = \mathbf{e}_x \cdot \mathbf{p} = \frac{\hbar}{i} \mathbf{e}_x \cdot \nabla , \qquad p_y = \mathbf{e}_y \cdot \mathbf{p} = \frac{\hbar}{i} \mathbf{e}_y \cdot \nabla , \qquad p_z = \mathbf{e}_z \cdot \mathbf{p} = \frac{\hbar}{i} \mathbf{e}_z \cdot \nabla .$$

and

$$p_\rho = \mathbf{e}_\rho \cdot \mathbf{p} = \frac{\hbar}{i} \mathbf{e}_\rho \cdot \nabla . \qquad p_\phi = \mathbf{e}_\phi \cdot \mathbf{p} = \frac{\hbar}{i} \mathbf{e}_\phi \cdot \nabla , \qquad p_z = \mathbf{e}_z \cdot \mathbf{p} = \frac{\hbar}{i} \mathbf{e}_z \cdot \nabla$$

3. Mathematica program for differential operators in cylindrical coordinates

The following Mathematica program (combination of #, &, pure function) is made by us for calculating differential operators in the cylindrical coordinates.

((**Mathematica**))

Here we use the following notations in the Mathematica program for the cylindrical coordinates.

$$r \to \sqrt{\rho^2 + z^2} \,,$$

$$u_x \to \mathbf{e}_x = \cos\phi \ \mathbf{e}_\rho - \sin\phi \ \mathbf{e}_\phi \,,$$

$$u_y \to \mathbf{e}_y = \sin\phi \ \mathbf{e}_\rho + \cos\phi \ \mathbf{e}_\phi \,,$$

$$u_z \to \ \mathbf{e}_z \,,$$

$$\mathbf{r} \to \rho \ \mathbf{e}_\rho + z \ \mathbf{e}_z \,,$$

$$u_r = \frac{\mathbf{r}}{r} \to \frac{1}{r}(\rho \ \mathbf{e}_\rho + z \ \mathbf{e}_z) \,,$$

$$x \to \mathbf{e}_x \cdot \mathbf{r} = \rho\cos\phi \,,$$

$$y \to \mathbf{e}_y \cdot \mathbf{r} = \rho\sin\phi \,,$$

$$z \to \mathbf{e}_z \cdot \mathbf{r} = z \,,$$

$$\mathrm{Lap} \to \nabla^2 \,,$$

$$\mathrm{Gra} \to grad, \ \ \nabla \,,$$

$$\mathrm{Diva} \to div \,,$$

$$\mathrm{Curla} \to curl \quad \text{or} \quad rot$$

$$L \to \mathbf{L} = \mathbf{r} \times \mathbf{p} \,,$$

$$L_x \to \mathbf{e}_x \cdot \mathbf{L}$$

$$L_y \to \mathbf{e}_y \cdot \mathbf{L}$$

$$L_z \to \mathbf{e}_z \cdot \mathbf{L}$$

$$LP \to L_x + iL_y$$

$$LM \to L_x - iL_y$$

$$Leq \to \mathbf{L}^2 = L_x{}^2 + L_y{}^2 + L_z{}^2$$

$$p \to \mathbf{p} = \frac{\hbar}{i}\nabla$$

$$p_x \to \mathbf{e}_x \cdot \mathbf{p} = \frac{\hbar}{i}\mathbf{e}_x \cdot \nabla$$

$$p_y \to \mathbf{e}_y \cdot \mathbf{p} = \frac{\hbar}{i}\mathbf{e}_y \cdot \nabla \,.$$

$$p_z \to \mathbf{e}_z \cdot \mathbf{p} = \frac{\hbar}{i}\mathbf{e}_z \cdot \nabla \,.$$

$$Peq \to \mathbf{p}^2 = p_x{}^2 + p_y{}^2 + p_z{}^2 \,.$$

$$prq \rightarrow \frac{1}{2}(\mathbf{e}_r \cdot \mathbf{p} + \mathbf{p} \cdot \mathbf{e}_r) \qquad \text{(radial linear momentum).}$$

$$p_\rho \rightarrow \mathbf{e}_\rho \cdot \mathbf{p}$$

((**Mathematica**))

Vector analysis
Angular momentum in the cylindrical coordinates

```
Clear["Global`"];
r1 = Sqrt[z^2 + ρ^2];
ux = {Cos[ϕ], -Sin[ϕ], 0};
uy = { Sin[ϕ], Cos[ϕ], 0};
uz = {0, 0, 1};
r = {ρ, 0, z};
ur = 1/r1 {ρ, 0, z};
Gra := Grad[#, {ρ, ϕ, z}, "Cylindrical"] &;
Lap := Laplacian[#, {ρ, ϕ, z}, "Cylindrical"] &;
Curla := Curl[#, {ρ, ϕ, z}, "Cylindrical"] &;
Diva := Div[#, {ρ, ϕ, z}, "Cylindrical"] &;
L := (-ⅈ ℏ Cross[r, Gra[#]]) &;
Lx := (ux.L[#] &) // Simplify;
Ly := (uy.L[#] &) // Simplify;
Lz := (uz.L[#] &) // Simplify;
LP := (Lx[#] + ⅈ Ly[#]) &;
LM := (Lx[#] - ⅈ Ly[#]) &;
prq := ((-ⅈ ℏ)/2 ur .Gra[#] + (-ⅈ ℏ)/2 Diva[# ur]) &;
Leq := (Nest[Lx, #, 2] + Nest[Ly, #, 2] + Nest[Lz, #, 2]) &;
p := (-ⅈ ℏ Gra[#]) &;
px := (ux.p[#]) &;
py := (uy.p[#]) &;
pz := (uz.p[#]) &;
pρ := ({1, 0, 0}.p[#]) &;
```

4. Expressions of Laplacian ∇^2, Gradient, Divergence, and Curl (Rotation) in the cylindrical co-ordinates

$$\nabla^2\psi(\rho,\phi,z) = \frac{\partial^2}{\partial\rho^2}\psi(\rho,\phi,z) + \frac{1}{\rho}\frac{\partial}{\partial\rho}\psi(\rho,\phi,z)$$
$$+\frac{1}{\rho^2}\frac{\partial^2}{\partial\phi^2}\psi(\rho,\phi,z) + \frac{\partial^2}{\partial z^2}\psi(\rho,\phi,z)$$
$$= \frac{1}{\rho}\frac{\partial}{\partial\rho}[\rho\frac{\partial}{\partial\rho}\psi(\rho,\phi,z)] + \frac{1}{\rho^2}\frac{\partial^2}{\partial\phi^2}\psi(\rho,\phi,z) + \frac{\partial^2}{\partial z^2}\psi(\rho,\phi,z)$$

$$\nabla\psi(\rho,\phi,z) = \mathbf{e}_\rho\frac{\partial}{\partial\rho}\psi(\rho,\phi,z) + \mathbf{e}_\phi\frac{1}{\rho}\frac{\partial}{\partial\phi}\psi(\rho,\phi,z) + \mathbf{e}_z\frac{\partial}{\partial z}\psi(\rho,\phi,z)$$

$$\nabla\cdot\mathbf{V}(\rho,\phi,z) = \frac{\partial}{\partial\rho}V_\rho(\rho,\phi,z) + \frac{1}{\rho}V_\rho(\rho,\phi,z) + \frac{1}{\rho}\frac{\partial}{\partial\phi}V_\phi(\rho,\phi,z) + \frac{\partial}{\partial z}V_z(\rho,\phi,z)$$
$$= \frac{1}{\rho}\frac{\partial}{\partial\rho}[\rho V_\rho(\rho,\phi,z)] + \frac{1}{\rho}\frac{\partial}{\partial\phi}V_\phi(\rho,\phi,z) + \frac{\partial}{\partial z}V_z(\rho,\phi,z)$$

$$\nabla\times\mathbf{V}(\rho,\phi,z) = \frac{1}{\rho}\mathbf{e}_\rho[\frac{\partial}{\partial\phi}V_z(\rho,\phi,z) - \frac{\partial}{\partial z}\rho V_\phi(\rho,\phi,z)]$$
$$-\frac{1}{\rho}(\rho\mathbf{e}_\phi)[\frac{\partial}{\partial\rho}V_z(\rho,\phi,z) - \frac{\partial}{\partial z}V_\rho(\rho,\phi,z)]$$
$$+\frac{1}{\rho}\mathbf{e}_z[\frac{\partial}{\partial\rho}\rho V_\phi(\rho,\phi,z) - \frac{\partial}{\partial\phi}V_\rho(\rho,\phi,z)]$$

or

$$\nabla\times\mathbf{V}(\rho,\phi,z) = \frac{1}{\rho}\begin{vmatrix} \mathbf{e}_\rho & \rho\mathbf{e}_\phi & \mathbf{e}_z \\ \frac{\partial}{\partial\rho} & \frac{\partial}{\partial\phi} & \frac{\partial}{\partial z} \\ V_\rho(\rho,\phi,z) & \rho V_\phi(\rho,\phi,z) & V_z(\rho,\phi,z) \end{vmatrix}$$

((**Mathematica**))

Laplacian, grad, div, and rot (curl)

```
Lap[ψ[ρ, ϕ, z]] // Simplify
```

$$\psi^{(0,0,2)}[\rho,\phi,z]+\frac{\psi^{(0,2,0)}[\rho,\phi,z]}{\rho^2}+\frac{\psi^{(1,0,0)}[\rho,\phi,z]}{\rho}+\psi^{(2,0,0)}[\rho,\phi,z]$$

```
Gra[ψ[ρ, ϕ, z]] // Simplify
```

$$\left\{\psi^{(1,0,0)}[\rho,\phi,z],\ \frac{\psi^{(0,1,0)}[\rho,\phi,z]}{\rho},\ \psi^{(0,0,1)}[\rho,\phi,z]\right\}$$

```
Diva[{Vρ[ρ, ϕ, z], Vϕ[ρ, ϕ, z], Vz[ρ, ϕ, z]}]
```

$$\mathrm{Vz}^{(0,0,1)}[\rho,\phi,z]+\frac{\mathrm{V\rho}[\rho,\phi,z]+\mathrm{V\phi}^{(0,1,0)}[\rho,\phi,z]}{\rho}+\mathrm{V\rho}^{(1,0,0)}[\rho,\phi,z]$$

```
Curla[{Vρ[ρ, ϕ, z], Vϕ[ρ, ϕ, z], Vz[ρ, ϕ, z]}] //
 Simplify
```

$$\left\{-\mathrm{V\phi}^{(0,0,1)}[\rho,\phi,z]+\frac{\mathrm{Vz}^{(0,1,0)}[\rho,\phi,z]}{\rho},\ \mathrm{V\rho}^{(0,0,1)}[\rho,\phi,z]-\mathrm{Vz}^{(1,0,0)}[\rho,\phi,z],\ \frac{\mathrm{V\phi}[\rho,\phi,z]-\mathrm{V\rho}^{(0,1,0)}[\rho,\phi,z]+\rho\,\mathrm{V\phi}^{(1,0,0)}[\rho,\phi,z]}{\rho}\right\}$$

5. Expressions of the orbital angular momentum in the cylindrical coordinates

We discuss the differential operators in the cylindrical coordinates,

$$L_x,\ L_y,\ L_z\ \ L_x \pm iL_y,\ \boldsymbol{L}^2 = L_x{}^2 + L_y{}^2 + L_z{}^2$$

in the cylindrical coordinates. The results are as follows.

(a)

$$L_z = \frac{\hbar}{i}\frac{\partial}{\partial\phi}$$

(b)

$$L_x = i\hbar(z\sin\phi\frac{\partial}{\partial\rho} + \frac{z}{\rho}\cos\phi\frac{\partial}{\partial\phi} - \rho\sin\phi\frac{\partial}{\partial z})$$

(c)

$$L_y = i\hbar(-z\cos\phi\frac{\partial}{\partial\rho} + \frac{z}{\rho}\sin\phi\frac{\partial}{\partial\phi} + \rho\cos\phi\frac{\partial}{\partial z})$$

The ladder operators for angular momentum,

$$\begin{aligned}
L_+ &= L_x + iL_y \\
&= i\hbar(z\sin\phi\frac{\partial}{\partial\rho} + \frac{z}{\rho}\cos\phi\frac{\partial}{\partial\phi} - \rho\sin\phi\frac{\partial}{\partial z}) \\
&\quad + i\hbar(-iz\cos\phi\frac{\partial}{\partial\rho} + \frac{z}{\rho}i\sin\phi\frac{\partial}{\partial\phi} + i\rho\cos\phi\frac{\partial}{\partial z}) \\
&= i\hbar e^{i\phi}(-iz\frac{\partial}{\partial\rho} + i\rho\frac{\partial}{\partial z} + \frac{z}{\rho}\frac{\partial}{\partial\phi})
\end{aligned}$$

$$\begin{aligned}
L_- &= L_x - iL_y \\
&= i\hbar(z\sin\phi\frac{\partial}{\partial\rho} + \frac{z}{\rho}\cos\phi\frac{\partial}{\partial\phi} - \rho\sin\phi\frac{\partial}{\partial z}) \\
&\quad + i\hbar(iz\cos\phi\frac{\partial}{\partial\rho} - \frac{z}{\rho}i\sin\phi\frac{\partial}{\partial\phi} - i\rho\cos\phi\frac{\partial}{\partial z}) \\
&= i\hbar e^{-i\phi}(iz\frac{\partial}{\partial\rho} - i\rho\frac{\partial}{\partial z} + \frac{z}{\rho}\frac{\partial}{\partial\phi})
\end{aligned}$$

((**Mathematica**))

Angular momentum in the cylindrical coordinate

L, Lx, Ly, Lz, Lx+iLy, Lx-iLy

```
L[ψ[ρ, ϕ, z]] // Simplify
```

$$\left\{\frac{i\, z\, \hbar\, \psi^{(0,1,0)}[\rho, \phi, z]}{\rho},\; i\, \hbar\, \left(\rho\, \psi^{(0,0,1)}[\rho, \phi, z] - z\, \psi^{(1,0,0)}[\rho, \phi, z]\right),\; -i\, \hbar\, \psi^{(0,1,0)}[\rho, \phi, z]\right\}$$

```
Lx[ψ[ρ, ϕ, z]] // FullSimplify
```

$$i\, \hbar\, \left(-\rho\, \text{Sin}[\phi]\, \psi^{(0,0,1)}[\rho, \phi, z] + \frac{z\, \text{Cos}[\phi]\, \psi^{(0,1,0)}[\rho, \phi, z]}{\rho} + z\, \text{Sin}[\phi]\, \psi^{(1,0,0)}[\rho, \phi, z]\right)$$

```
Ly[ψ[ρ, ϕ, z]] // FullSimplify
```

$$\frac{1}{\rho}\, i\, \hbar\, \left(z\, \text{Sin}[\phi]\, \psi^{(0,1,0)}[\rho, \phi, z] + \rho\, \text{Cos}[\phi]\, \left(\rho\, \psi^{(0,0,1)}[\rho, \phi, z] - z\, \psi^{(1,0,0)}[\rho, \phi, z]\right)\right)$$

```
Lx[ψ[ρ, ϕ, z]] + i Ly[ψ[ρ, ϕ, z]] // FullSimplify
```

$$-\frac{e^{i\,\phi}\, \hbar\, \left(\rho^2\, \psi^{(0,0,1)}[\rho, \phi, z] - i\, z\, \psi^{(0,1,0)}[\rho, \phi, z] - z\, \rho\, \psi^{(1,0,0)}[\rho, \phi, z]\right)}{\rho}$$

```
Lx[ψ[ρ, ϕ, z]] - i Ly[ψ[ρ, ϕ, z]] // FullSimplify
```

$$\frac{e^{-i\,\phi}\, \hbar\, \left(\rho^2\, \psi^{(0,0,1)}[\rho, \phi, z] + i\, z\, \psi^{(0,1,0)}[\rho, \phi, z] - z\, \rho\, \psi^{(1,0,0)}[\rho, \phi, z]\right)}{\rho}$$

```
Lz[ψ[ρ, ϕ, z]] // Simplify
```

$$-i\, \hbar\, \psi^{(0,1,0)}[\rho, \phi, z]$$

6. Commutation relations of anguylar momentum

Next, we discuss the commutation relations for the angular momentum using Mathematica.

$$[L_x, L_y] = i\hbar L_z, \qquad [L_y, L_z] = i\hbar L_x, \qquad [L_z, L_z] = i\hbar L_y$$

$$[\boldsymbol{L}^2, L_x] = 0, \qquad [\boldsymbol{L}^2, L_y] = 0, \qquad [\boldsymbol{L}^2, L_z] = 0$$

$$[L_z, L_+] = \hbar L_+ \qquad [L_z, L_-] = -\hbar L_-$$

((**Mathematica**))

```
Lx[Ly[ψ[ρ, ϕ, z]]] - Ly[Lx[ψ[ρ, ϕ, z]]] -
  ⅈ ℏ Lz[ψ[ρ, ϕ, z]] // Simplify

0

Ly[Lz[ψ[ρ, ϕ, z]]] - Lz[Ly[ψ[ρ, ϕ, z]]] -
  ⅈ ℏ Lx[ψ[ρ, ϕ, z]] // Simplify

0

Lz[Lx[ψ[ρ, ϕ, z]]] - Lx[Lz[ψ[ρ, ϕ, z]]] -
  ⅈ ℏ Ly[ψ[ρ, ϕ, z]] // Simplify // Simplify

0
```

```
Leq[Lx[ψ[ρ, ϕ, z]]] - Lx[Leq[ψ[ρ, ϕ, z]]] // Simplify
```

```
0
```

```
Leq[Ly[ψ[ρ, ϕ, z]]] - Ly[Leq[ψ[ρ, ϕ, z]]] // Simplify
```

```
0
```

```
Leq[Lz[ψ[ρ, ϕ, z]]] - Lz[Leq[ψ[ρ, ϕ, z]]] // Simplify
```

```
0
```

```
Lz[LP[ψ[ρ, ϕ, z]]] - LP[Lz[ψ[ρ, ϕ, z]]] - ℏ LP[ψ[ρ, ϕ, z]] //
 Simplify
```

```
0
```

```
Lz[LM[ψ[ρ, ϕ, z]]] - LM[Lz[ψ[ρ, ϕ, z]]] + ℏ LM[ψ[ρ, ϕ, z]] //
 Simplify
```

```
0
```

7. Expression of $L^2 = L_x{}^2 + L_y{}^2 + L_z{}^2$ in the circular cylindrical coordinates.

The square of the magnitude of total angular momentum is obtained as

$$\begin{aligned}
\mathbf{L}^2 &= L_x{}^2 + L_y{}^2 + L_z{}^2 \\
&= -\frac{\hbar^2}{\rho^2}\{-2z\rho^2\frac{\partial}{\partial z} + \rho^4\frac{\partial^2}{\partial z^2} + (z^2+\rho^2)\frac{\partial^2}{\partial \phi^2} \\
&\quad + z^2\rho\frac{\partial}{\partial \rho} - \rho^3\frac{\partial}{\partial \rho} + 2z\rho^3\frac{\partial}{\partial \rho}\frac{\partial}{\partial z} + z^2\rho^2\frac{\partial^2}{\partial \rho^2})
\end{aligned}$$

((**Mathematica**))

L^2 in the cylindrical coordinate

```
eq1 = Lx[Lx[ψ[ρ, ϕ, z]]] + Ly[Ly[ψ[ρ, ϕ, z]]] +
  Lz[ Lz[ψ[ρ, ϕ, z]]] // Simplify
```

$$-\frac{1}{\rho^2}\hbar^2\left(-2 z \rho^2 \psi^{(0,0,1)}[\rho,\phi,z]+\rho^4 \psi^{(0,0,2)}[\rho,\phi,z]+ z^2 \psi^{(0,2,0)}[\rho,\phi,z]+\rho^2 \psi^{(0,2,0)}[\rho,\phi,z]+ z^2 \rho \psi^{(1,0,0)}[\rho,\phi,z]-\rho^3 \psi^{(1,0,0)}[\rho,\phi,z]- 2 z \rho^3 \psi^{(1,0,1)}[\rho,\phi,z]+z^2 \rho^2 \psi^{(2,0,0)}[\rho,\phi,z]\right)$$

```
eq2 = Leq[ψ[ρ, ϕ, z]] // Simplify
```

$$-\frac{1}{\rho^2}\hbar^2\left(-2 z \rho^2 \psi^{(0,0,1)}[\rho,\phi,z]+\rho^4 \psi^{(0,0,2)}[\rho,\phi,z]+ z^2 \psi^{(0,2,0)}[\rho,\phi,z]+\rho^2 \psi^{(0,2,0)}[\rho,\phi,z]+ z^2 \rho \psi^{(1,0,0)}[\rho,\phi,z]-\rho^3 \psi^{(1,0,0)}[\rho,\phi,z]- 2 z \rho^3 \psi^{(1,0,1)}[\rho,\phi,z]+z^2 \rho^2 \psi^{(2,0,0)}[\rho,\phi,z]\right)$$

```
eq12 = eq1 - eq2 // Simplify
0
```

8. Radial linear momentum

The radial linear momentum is defined by

$$p_r = \frac{1}{2}\left(\frac{\mathbf{r}}{r}\cdot\mathbf{p}+\mathbf{p}\cdot\frac{\mathbf{r}}{r}\right)=\frac{\hbar}{2i}\left(\frac{\mathbf{r}}{r}\cdot\nabla+\nabla\cdot\frac{\mathbf{r}}{r}\right)$$

Using the Mathematica, the corresponding differential operator is obtained as

$$p_r = \frac{\hbar}{i}\frac{1}{\sqrt{\rho^2+z^2}}\left(1+z\frac{\partial}{\partial z}+\rho\frac{\partial}{\partial \rho}\right)$$

Note that

$$[p_r, r] = \frac{\hbar}{i}.$$

((**Mathematica**))

Radial momentum (quantum)

```
prq[ ψ[ρ, ϕ, z]] // Simplify
```

$$-\frac{i\hbar\left(\psi[\rho,\phi,z]+z\,\psi^{(0,0,1)}[\rho,\phi,z]+\rho\,\psi^{(1,0,0)}[\rho,\phi,z]\right)}{\sqrt{z^2+\rho^2}}$$

Commutation relation for radial momentum and magnitude of r

```
prq[r1 ψ[ρ, ϕ, z]] - r1 prq[ψ[ρ, ϕ, z]] // Simplify
```

$$-i\hbar\,\psi[\rho,\phi,z]$$

9. Expression of $\mathbf{p}^2$

We show that the differential operator $\mathbf{p}^2$ is expressed by

$$\mathbf{p}^2 = p_r^{\ 2} + \frac{\mathbf{L}^2}{z^2+\rho^2}$$

using Mathematica. The expression is so complicated that we expect. However, the expression is mathematically correct.

$$\mathbf{p}^2 = -\frac{\hbar^2}{\rho^2}\left(\rho^2\frac{\partial^2}{\partial z^2} + \frac{\partial^2}{\partial\phi^2} + \rho\frac{\partial}{\partial\rho} + \rho^2\frac{\partial^2}{\partial\rho^2}\right),$$

$$p_r^{\ 2} = -\frac{\hbar^2}{z^2+\rho^2}\left(2z\frac{\partial}{\partial z} + z^2\frac{\partial^2}{\partial z^2} + 2\rho\frac{\partial}{\partial\rho} + 2\rho z\frac{\partial^2}{\partial\rho\partial z} + \rho^2\frac{\partial^2}{\partial\rho^2}\right),$$

Thus, we have

$$\mathbf{p}^2 - p_r{}^2 = -\frac{\hbar^2}{\rho^2}(\rho^2 \frac{\partial^2}{\partial z^2} + \frac{\partial^2}{\partial \phi^2} + \rho \frac{\partial}{\partial \rho} + \rho^2 \frac{\partial^2}{\partial \rho^2})$$
$$+ \frac{\hbar^2}{z^2 + \rho^2}(2z \frac{\partial}{\partial z} + z^2 \frac{\partial^2}{\partial z^2} + 2\rho \frac{\partial}{\partial \rho} + 2\rho z \frac{\partial^2}{\partial \rho \partial z} + \rho^2 \frac{\partial^2}{\partial \rho^2})$$

So that we get

$$(\mathbf{p}^2 - p_r{}^2)(\rho^2 + z^2) = \mathbf{L}^2$$
$$= -\frac{\hbar^2}{\rho^2}[-2z\rho^2 \frac{\partial}{\partial z} + \rho^4 \frac{\partial^2}{\partial z^2} + (z^2 + \rho^2)\frac{\partial^2}{\partial \phi^2}$$
$$+ z^2 \rho \frac{\partial}{\partial \rho} - \rho^3 \frac{\partial}{\partial \rho} + 2z\rho^3 \frac{\partial}{\partial \rho}\frac{\partial}{\partial z} + z^2 \rho^2 \frac{\partial^2}{\partial \rho^2}]$$

or

$$\mathbf{p}^2 = p_r{}^2 + \frac{\mathbf{L}^2}{\rho^2 + z^2} = p_r{}^2 + \frac{\mathbf{L}^2}{r^2}$$

((**Mathematica**))

p^2 in the cylindrical coordinate

```
eq1 = px[px[ψ[ρ, ϕ, z]]] + py[py[ψ[ρ, ϕ, z]]] +
   pz[ pz[ψ[ρ, ϕ, z]]] // Simplify
```

$$-\frac{1}{\rho^2}\hbar^2 \left(\rho^2 \psi^{(0,0,2)}[\rho, \phi, z] + \psi^{(0,2,0)}[\rho, \phi, z] + \rho \left(\psi^{(1,0,0)}[\rho, \phi, z] + \rho\, \psi^{(2,0,0)}[\rho, \phi, z]\right)\right)$$

pr^2 in the cylindrical coordinate

```
eq2 = prq[prq[ψ[ρ, ϕ, z]]] // Simplify
```

$$-\frac{1}{z^2+\rho^2}\hbar^2 \left(2\, z\, \psi^{(0,0,1)}[\rho, \phi, z] + z^2 \psi^{(0,0,2)}[\rho, \phi, z] + \rho \left(2\, \psi^{(1,0,0)}[\rho, \phi, z] + 2\, z\, \psi^{(1,0,1)}[\rho, \phi, z] + \rho\, \psi^{(2,0,0)}[\rho, \phi, z]\right)\right)$$

```
eq12 = (eq1 - eq2) (z^2 + ρ^2) // Simplify
```

$$-\frac{1}{\rho^2}\hbar^2 \left(-2\, z\, \rho^2\, \psi^{(0,0,1)}[\rho, \phi, z] + \rho^4\, \psi^{(0,0,2)}[\rho, \phi, z] + z^2\, \psi^{(0,2,0)}[\rho, \phi, z] + \rho^2\, \psi^{(0,2,0)}[\rho, \phi, z] + z^2\, \rho\, \psi^{(1,0,0)}[\rho, \phi, z] - \rho^3\, \psi^{(1,0,0)}[\rho, \phi, z] - 2\, z\, \rho^3\, \psi^{(1,0,1)}[\rho, \phi, z] + z^2\, \rho^2\, \psi^{(2,0,0)}[\rho, \phi, z]\right)$$

L^2 in the cylindrical coordinates

```
eq3 = Leq[ψ[ρ, ϕ, z]] // Simplify
```

$$-\frac{1}{\rho^2}\hbar^2 \left(-2\, z\, \rho^2\, \psi^{(0,0,1)}[\rho, \phi, z] + \rho^4\, \psi^{(0,0,2)}[\rho, \phi, z] + z^2\, \psi^{(0,2,0)}[\rho, \phi, z] + \rho^2\, \psi^{(0,2,0)}[\rho, \phi, z] + z^2\, \rho\, \psi^{(1,0,0)}[\rho, \phi, z] - \rho^3\, \psi^{(1,0,0)}[\rho, \phi, z] - 2\, z\, \rho^3\, \psi^{(1,0,1)}[\rho, \phi, z] + z^2\, \rho^2\, \psi^{(2,0,0)}[\rho, \phi, z]\right)$$

```
eq4 = eq12 - eq3 // Simplify
```

0

10. Example-1 (Goswami, Quantum Mechanics p.265 problem-4)

Prove the following equivalence:

$$p_r r^2 p_r = r p_r^{\,2} r$$

with

$$[p_r, r] = -i\hbar$$

((**Proof**))

We use the commutation relation $\left[p_r, r\right] = -i\hbar$.

$$p_r r^2 p_r = (p_r r)(r p_r)$$
$$= (r p_r - i\hbar)(p_r r + i\hbar)$$
$$= (r p_r p_r r) + i\hbar(r p_r - p_r r) + \hbar^2$$
$$= r p_r^{\ 2} r$$

((**Mathematica**))

```
prq[r1 ψ[ρ, ϕ, z]] - r1 prq[ψ[ρ, ϕ, z]] //
 Simplify

-ⅈ ℏ ψ[ρ, ϕ, z]

prq[r1^2 prq[ψ[ρ, ϕ, z]]] -
  r1 prq[ prq[r1 ψ[ρ, ϕ, z]]] // Simplify

0
```

11. Example-2: Goswami Quantum Mechanics, P.265 Problem 5

((**Problem-5**)) Find the expression for ∇^2 in the cylindrical coordinates. Then show that

$$\rho^2 \boldsymbol{p}^2 = (\rho p_\rho)^2 + L_z^{\ 2} + \rho^2 p_z^{\ 2}$$

where

$$p_\rho = \mathbf{e}_\rho \cdot \mathbf{p} = \frac{\hbar}{i}\frac{\partial}{\partial \rho}, \qquad L_z = \frac{\hbar}{i}\frac{\partial}{\partial \phi}.$$

$$p_z = \frac{\hbar}{i}\frac{\partial}{\partial z}.$$

$$p_x = \frac{\hbar}{i\rho}(\rho \cos\phi \frac{\partial}{\partial \rho} - \sin\phi \frac{\partial}{\partial \phi}),$$

$$p_y = \frac{\hbar}{i\rho}(\rho \sin\phi \frac{\partial}{\partial \rho} + \cos\phi \frac{\partial}{\partial \phi})$$

Using Mathematica we get

$$\rho^2 \mathbf{p}^2 = -\hbar^2 (\rho^2 \frac{\partial^2}{\partial z^2} + \frac{\partial^2}{\partial \phi^2} + \rho^2 \frac{\partial^2}{\partial \rho^2} + \rho \frac{\partial}{\partial \rho})$$

$$(\rho p_\rho)^2 = \rho p_\rho \rho p_\rho = -\hbar^2 (\rho \frac{\partial}{\partial \rho} + \rho^2 \frac{\partial^2}{\partial \rho^2})$$

$$L_z^{\ 2} = -\hbar^2 \frac{\partial^2}{\partial \phi^2}$$

$$\rho^2 p_z^{\ 2} = -\hbar^2 \rho^2 \frac{\partial^2}{\partial z^2}$$

$$\begin{aligned} \rho^2 \boldsymbol{p}^2 &= (\rho p_\rho)^2 + L_z^{\ 2} + \rho^2 p_z^{\ 2} \\ &= -\hbar^2 (\rho^2 \frac{\partial^2}{\partial \rho^2} + \rho \frac{\partial}{\partial \rho} + \frac{\partial^2}{\partial \phi^2} + \rho^2 \frac{\partial^2}{\partial z^2}) \end{aligned}$$

or

$$\begin{aligned} \boldsymbol{p}^2 &= -\hbar^2 (\frac{\partial^2}{\partial \rho^2} + \frac{1}{\rho} \frac{\partial}{\partial \rho} + \frac{1}{\rho^2} \frac{\partial^2}{\partial \phi^2} + \frac{\partial^2}{\partial z^2}) \\ &= -\hbar^2 (\frac{1}{\rho} \frac{\partial}{\partial \rho} \rho \frac{\partial}{\partial \rho} + \frac{1}{\rho^2} \frac{\partial^2}{\partial \phi^2} + \frac{\partial^2}{\partial z^2}) \end{aligned}$$

Note that

$$\frac{\partial^2}{\partial \rho^2} + \frac{1}{\rho} \frac{\partial}{\partial \rho} = \frac{1}{\rho} \frac{\partial}{\partial \rho} (\rho \frac{\partial}{\partial \rho}).$$

((**Mathematica**))

```
eq1 =
 ρ^2 (Nest[px, ψ[ρ, ϕ, z], 2] + Nest[py, ψ[ρ, ϕ, z], 2] +
     Nest[pz, ψ[ρ, ϕ, z], 2]) // Simplify
```

$$-\hbar^2 \left(\rho^2 \psi^{(0,0,2)}[\rho, \phi, z] + \psi^{(0,2,0)}[\rho, \phi, z] + \rho \left(\psi^{(1,0,0)}[\rho, \phi, z] + \rho\, \psi^{(2,0,0)}[\rho, \phi, z]\right)\right)$$

```
eq2 = ρ pρ[ρ pρ[ψ[ρ, ϕ, z]]] // FullSimplify
```

$$-\rho\, \hbar^2 \left(\psi^{(1,0,0)}[\rho, \phi, z] + \rho\, \psi^{(2,0,0)}[\rho, \phi, z]\right)$$

```
eq3 = Lz[Lz[ψ[ρ, ϕ, z]]] // FullSimplify
```

$$-\hbar^2 \psi^{(0,2,0)}[\rho, \phi, z]$$

```
eq4 = ρ^2 pz[pz[ ψ[ρ, ϕ, z]]] // FullSimplify
```

$$-\rho^2 \hbar^2 \psi^{(0,0,2)}[\rho, \phi, z]$$

```
eq1 - (eq2 + eq3 + eq4) // Simplify
```

0

---

12. Example-3

Show that

$$[\mathbf{L}^2,[\mathbf{L}^2,x]] = 2\hbar^2(x\mathbf{L}^2 + \mathbf{L}^2 x)$$

With

$$x = \mathbf{e}_x \cdot \mathbf{r}$$

```
x1 = ux.r

ρ Cos[ϕ]

eq1 =
 Leq[Leq [x1 ψ[ρ, ϕ, z]]] - Leq[x1 Leq[ψ[ρ, ϕ, z]]] -
   Leq[x1 Leq[ψ[ρ, ϕ, z]]] + x1 Leq[Leq[ψ[ρ, ϕ, z]]] //
  FullSimplify

1/ρ 4 ℏ^4 (ρ^2 Cos[ϕ] ψ[ρ, ϕ, z] +
    (z^2 + ρ^2) Sin[ϕ] ψ^(0,1,0)[ρ, ϕ, z] -
    Cos[ϕ] (-3 z ρ^2 ψ^(0,0,1)[ρ, ϕ, z] +
       z^2 ψ^(0,2,0)[ρ, ϕ, z] + ρ (2 z^2 ψ^(1,0,0)[ρ, ϕ, z] +
          ρ (ψ^(0,2,0)[ρ, ϕ, z] + ρ (ρ ψ^(0,0,2)[ρ, ϕ, z] -
                ψ^(1,0,0)[ρ, ϕ, z] - 2 z ψ^(1,0,1)[ρ, ϕ, z]) +
             z^2 ψ^(2,0,0)[ρ, ϕ, z]))))

eq2 = 2 ℏ^2 (x1 Leq[ψ[ρ, ϕ, z]] + Leq[x1 ψ[ρ, ϕ, z]]) //
  FullSimplify

1/ρ 4 ℏ^4 (ρ^2 Cos[ϕ] ψ[ρ, ϕ, z] +
    (z^2 + ρ^2) Sin[ϕ] ψ^(0,1,0)[ρ, ϕ, z] -
    Cos[ϕ] (-3 z ρ^2 ψ^(0,0,1)[ρ, ϕ, z] +
       z^2 ψ^(0,2,0)[ρ, ϕ, z] + ρ (2 z^2 ψ^(1,0,0)[ρ, ϕ, z] +
          ρ (ψ^(0,2,0)[ρ, ϕ, z] + ρ (ρ ψ^(0,0,2)[ρ, ϕ, z] -
                ψ^(1,0,0)[ρ, ϕ, z] - 2 z ψ^(1,0,1)[ρ, ϕ, z]) +
             z^2 ψ^(2,0,0)[ρ, ϕ, z]))))

eq12 = eq1 - eq2 // Simplify

0
```

13. Example-4: Problem (9-23) Townsend, A Modern Approach to Quantum Mechanics

((**9-23**)) The Hamiltonian for a three-dimensional (3D) system with cylindrical symmetry is given by

$$\hat{H} = \frac{1}{2\mu}\hat{\mathbf{p}}^2 + V(\hat{\rho})$$

where $\quad |\hat{\rho}| = \sqrt{\hat{x}^2 + \hat{y}^2}$.

(a) Use symmetry arguments to establish that both $\hat{p}_z$, the generator of translations in the z direction, and $\hat{L}_z$, the generator of rotations about the z axis, commute with $\hat{H}$.

(b) Use the fact that $\hat{H}$, $\hat{p}_z$, and $\hat{L}_z$ have eigenstates in common to express the position-space eigenfunctions of the Hamiltonian in terms of those of $\hat{p}_z$ and $\hat{L}_z$.

(c) What is the radial equation? Note: The Laplacian in cylindrical coordinates is given by

$$\nabla^2\psi(\rho,\phi,z) = \frac{1}{\rho}\frac{\partial}{\partial\rho}[\rho\frac{\partial}{\partial\rho}\psi(\rho,\phi,z)] + \frac{1}{\rho^2}\frac{\partial^2}{\partial\phi^2}\psi(\rho,\phi,z) + \frac{\partial^2}{\partial z^2}\psi(\rho,\phi,z)$$

((**Solution**))

The Hamiltonian:

$$\hat{H} = \frac{1}{2\mu}\hat{\boldsymbol{p}}^2 + V(\hat{\rho})$$

The commutation relation:

$$\left[\hat{H},\hat{p}_z\right] = 0, \qquad \left[\hat{H},\hat{L}_z\right] = 0$$

Then, we have a simultaneous eigenket;

$$H\left|E,m,k_z\right\rangle = E\left|E,m,k_z\right\rangle, \qquad \hat{L}_z\left|E,m,k_z\right\rangle = m\hbar\left|E,m,k_z\right\rangle$$

$$\hat{p}_z\left|E,m,k_z\right\rangle = \hbar k_z\left|E,m,k_z\right\rangle$$

The wave function can be expressed as

$$\psi(\rho,\phi,k_z) = e^{im\phi}e^{ik_z z}\chi(\rho)$$

$\chi(\rho)$ satisfies the following differential equation,

$$-\frac{\hbar^2}{2\mu\rho}[\chi'(\rho)+\rho\chi''(\rho)]+[\frac{\hbar^2}{2\mu}(k_z{}^2+\frac{m^2}{\rho^2})+V(\rho)]\chi(\rho)=E\chi(\rho)$$

or

$$-\frac{\hbar^2}{2\mu\rho}\frac{d}{d\rho}(\rho\chi'(\rho)+[\frac{\hbar^2}{2\mu}(k_z{}^2+\frac{m^2}{\rho^2})+V(\rho)]\chi(\rho)=E\chi(\rho)$$

((**Mathematica**))

Hamiltonian

```
H1 := (1/(2 μ) Peq[#] + V1[ρ] #) &;
```

Schrodinger equation

```
eq11 = H1[ψ[ρ, ϕ, z]] - E1 ψ[ρ, ϕ, z] // Simplify
```

$$(-E1 + V1[\rho])\,\psi[\rho, \phi, z] - \frac{1}{2\,\mu\,\rho^2}\hbar^2\left(\rho^2\,\psi^{(0,0,2)}[\rho, \phi, z] + \psi^{(0,2,0)}[\rho, \phi, z] + \rho\left(\psi^{(1,0,0)}[\rho, \phi, z] + \rho\,\psi^{(2,0,0)}[\rho, \phi, z]\right)\right)$$

Commutation relation between H and Lz

```
eq12 = H1[Lz[ψ[ρ, ϕ, z]]] - Lz[H1[ψ[ρ, ϕ, z]]] // Simplify
```

0

```
eq21 = Lz[H1 [ψ[ρ, ϕ, z]]] // Simplify
```

$$-i \hbar V1[\rho]\, \psi^{(0,1,0)}[\rho, \phi, z] + \frac{1}{2 \mu \rho^2} i \hbar^3 \left(\rho^2 \psi^{(0,1,2)}[\rho, \phi, z] + \psi^{(0,3,0)}[\rho, \phi, z] + \rho \left(\psi^{(1,1,0)}[\rho, \phi, z] + \rho\, \psi^{(2,1,0)}[\rho, \phi, z]\right)\right)$$

Commutation relation between H and pz

```
eq22 = H1[pz [ψ[ρ, ϕ, z]]] - pz [ H1[ψ[ρ, ϕ, z]]] // Simplify
0
```

Simultaneous eigenket of pz and Lz

```
eq31 = H1[Lz[ψ[ρ, ϕ, z]]] - Lz[H1[ψ[ρ, ϕ, z]]] // Simplify
0
```

```
eq41 = Lz[ψ[ρ, ϕ, z]] == m ℏ ψ[ρ, ϕ, z] // Simplify
```

$$-i \hbar \psi^{(0,1,0)}[\rho, \phi, z] == m \hbar \psi[\rho, \phi, z]$$

```
eq42 = DSolve[eq41, ψ[ρ, ϕ, z], ϕ]
```

$$\{\{\psi[\rho, \phi, z] \to e^{i m \phi} c_1\}\}$$

```
eq51 = pz [ψ[ρ, ϕ, z]] == ℏ kz ψ[ρ, ϕ, z] // Simplify
```

$$-i \hbar \psi^{(0,0,1)}[\rho, \phi, z] == kz \hbar \psi[\rho, \phi, z]$$

```
eq52 = DSolve[eq51, ψ[ρ, ϕ, z], z] // Simplify
```

$$\{\{\psi[\rho, \phi, z] \to e^{i kz z} c_1\}\}$$

```
ψ[ρ, ϕ, z] /. eq52
```

$$\{e^{i kz z} c_1\}$$

```
eq41 = Lz[ψ[ρ, ϕ, z]] == m ℏ ψ[ρ, ϕ, z] // Simplify
```

$-i\hbar\,\psi^{(0,1,0)}[\rho, \phi, z] == m\,\hbar\,\psi[\rho, \phi, z]$

```
eq42 = DSolve[eq41, ψ[ρ, ϕ, z], ϕ]
```

$\{\{\psi[\rho, \phi, z] \to e^{i m \phi}\, c_1\}\}$

```
eq51 = pz [ψ[ρ, ϕ, z]] == ℏ kz ψ[ρ, ϕ, z] // Simplify
```

$-i\hbar\,\psi^{(0,0,1)}[\rho, \phi, z] == kz\,\hbar\,\psi[\rho, \phi, z]$

```
eq52 = DSolve[eq51, ψ[ρ, ϕ, z], z] // Simplify
```

$\{\{\psi[\rho, \phi, z] \to e^{i\, kz\, z}\, c_1\}\}$

```
ψ[ρ, ϕ, z] /. eq52
```

$\{e^{i\, kz\, z}\, c_1\}$

The form of the wave function

```
rule2 = {ψ → (χ[#1] Exp[i m #2] Exp[i kz #3] &)};
eq61 = eq11 /. rule2 // FullSimplify
```

$$\frac{1}{2\mu\rho^2} e^{i\,(kz\,z+m\,\phi)} \left(\left(-2\,E1\,\mu\,\rho^2 + \left(m^2 + kz^2\,\rho^2\right)\hbar^2 + 2\,\mu\,\rho^2\,V1[\rho]\right)\chi[\rho] - \rho\,\hbar^2\,(\chi'[\rho] + \rho\,\chi''[\rho])\right)$$

14. Example-5 (Townsend, Problem 10-11))

A particle of mass μ is in the cylindrical potential well

$$V(\rho) = \begin{cases} 0 & (\rho < a) \\ \infty & (\rho > a) \end{cases}$$

With $\rho = \sqrt{x^2 + y^2}$.

(a) Determine the three lowest energy eigenvalues for states that also have p_z and L_z equal to zero.

(b) Determine the three lowest energy eigenvalues for states with p_z equal to zero. The states may have nonzero L_z .

((**Solution**))

We start with the Schrodinger equation derived in the section 14.

$$\begin{aligned} -\frac{\hbar^2}{2\mu}\nabla^2\psi(\rho,\phi,z) + V(\rho)\psi(\rho,\phi,z) &= -\frac{\hbar^2}{2\mu}[\frac{1}{\rho}\frac{\partial}{\partial\rho}[\rho\frac{\partial}{\partial\rho}\psi(\rho,\phi,z)] + \frac{1}{\rho^2}\frac{\partial^2}{\partial\phi^2}\psi(\rho,\phi,z) \\ &\quad + \frac{\partial^2}{\partial z^2}\psi(\rho,\phi,z)] + V(\rho)\psi(\rho,\phi,z) \\ &= E\psi(\rho,\phi,z) \end{aligned}$$

or

$$\begin{aligned} &\rho^2\frac{\partial^2}{\partial z^2}\psi(\rho,\phi,z) + \frac{\partial^2}{\partial\phi^2}\psi(\rho,\phi,z) + \rho\frac{\partial}{\partial\rho}\psi(\rho,\phi,z)] \\ &+\rho^2\frac{\partial^2}{\partial\rho^2}\psi(\rho,\phi,z) - \frac{2\mu\rho^2}{\hbar^2}[V(\rho) - E]\psi(\rho,\phi,z) \\ &= 0 \end{aligned}$$

We assume that

$$\psi(\rho,\phi,k_z) = e^{im\phi}e^{ik_z z}\chi(\rho) \qquad \text{(separation of variables)}$$

or

$$\rho^2\frac{\partial^2\chi(\rho)}{\partial\rho^2} + \rho\frac{\partial\chi(\rho)}{\partial\rho} - \{(\rho^2 k_z^{\ 2} + m^2) + \frac{2\mu\rho^2}{\hbar^2}[V(\rho) - E]\}\chi(\rho) = 0$$

Using the relation

$$\rho^2 \frac{\partial^2 \chi(\rho)}{\partial \rho^2} + \rho \frac{\partial \chi(\rho)}{\partial \rho} = \rho \frac{\partial}{\partial \rho}[\rho \frac{\partial \chi(\rho)}{\partial \rho}]$$

or

$$\rho \frac{\partial}{\partial \rho}[\rho \frac{\partial \chi(\rho)}{\partial \rho}] - \{(\rho^2 k_z^{\ 2} + m^2) + \frac{2\mu\rho^2}{\hbar^2}[V(\rho) - E]\}\chi(\rho) = 0$$

We now consider the case of $V(\rho) = 0$ for $\rho < a$. We use

$$E = \frac{\hbar^2 k^2}{2\mu}.$$

$$\rho \frac{\partial}{\partial \rho}[\rho \frac{\partial \chi(\rho)}{\partial \rho}] + [\rho^2 (k^2 - k_z^{\ 2}) - m^2]\chi(\rho) = 0$$

Here we have a change of variable such that $\xi = \sqrt{k^2 - k_z^{\ 2}}\rho = \alpha\rho$

$$\xi \frac{\partial}{\partial \xi}[\xi \frac{\partial \chi(\xi)}{\partial \xi}] + [\xi^2 - m^2]\chi(\xi) = 0.$$

The solution of this Bessel differential equation is obtained as

$$\chi(\xi) = AJ_m(\xi).$$

The energy eigenvalues are derived from the zero points of the Bessel function,

$$J_m(\sqrt{k^2 - k_z^{\ 2}}a) = 0. \qquad \text{with } m = 0, 1, 2, 3, \ldots.$$

Note that the Bessel function $J_m(x)$ becomes equal to zero at $\xi = x = x(m,n)$, where n = 1, 2, 3, 4,.......

$$\sqrt{k^2 - k_z^{\ 2}}a = x(m,n),$$

$$E(m,n) = \frac{\hbar^2 k^2}{2\mu} = \frac{\hbar^2 k_z^{\ 2}}{2\mu} + \frac{\hbar^2}{2\mu a^2}[x(m,n)]^2$$



Fig. Energy eigenvalues $\frac{E(m,n)}{E_0} = [x(m,n)]^2$ for the state {m,n}, with $k_z = 0$ and

$E_0 = \frac{\hbar^2}{2\mu a^2}$.

Table:

The value of $x(m,n)$ where that the Bessel function $J_m(x)$ becomes equal to zero, where n = 1, 2, 3, 4,.......

n	$J_0(x)$	$J_1(x)$	$J_2(x)$	$J_3(x)$
1	2.40483	3.83171	5.13562	6.38016
2	5.52008	7.01559	8.41724	9.76102
3	8.65373	10.1735	11.6198	13.0152
4	11.7915	13.3237	14.796	16.2235
5	14.9309	16.4706	17.9598	19.4094
6	18.0711	19.6159	21.117	22.5827
7	21.2116	22.7601	24.2701	25.7482
8	24.3525	25.9037	27.4206	28.9084
9	27.4935	29.0468	30.5692	32.0649
10	30.6346	32.1897	33.7165	35.2187

---

n	$J_4(x)$	$J_5(x)$	$J_6(x)$	$J_7(x)$
1	7.58834	8.77148	9.93611	11.0864
2	11.0647	12.3386	13.5893	14.8213
3	14.3725	15.7002	17.0038	18.2876
4	17.616	18.9801	20.3208	21.6415
5	20.8269	22.2178	23.5861	24.9349
6	24.019	25.4303	26.8202	28.1912
7	27.1991	28.6266	30.0337	31.4228
8	30.371	31.8117	33.233	34.6371
9	33.5371	34.9888	36.422	37.8387
10	36.699	38.1599	39.6032	41.0308



Fig. Plot of the Bessel function $J_m(x)$ as a function of x, with m = 0, 1, 2, 3, 4,....

15. Summary

The differential operator in the Mathematica can be made with the Function function. Using the above Mathematica program, one may solve more complicated problems in quantum mechanics (circular cylindrical coordinates).

REFERENCES

J.S. Townsend, *A Modern Approach to Quantum Mechanics*, 2nd edition (University Science Book, 2012).

A. Goswami, *Quantum Mechanics*, 2nd edition (Waveland, 2003)

J. Schwinger, Quantum Mechanics Symbolism of Atomic Measurements (Springer, 2001).

J.J. Sakurai and J. Napolitano, Modern Quantum Mechanics, 3rd edition , edited by B.-G. Englert (Cambridge, 2021).

G.B. Arfken and H.J. Weber, Mathematical Methods for Physicists, 5th edition (Harcourt Academic Press, 2001).

R.L. Zimmerman and F.I. Olness, Mathematica for Physics book, 2nd edition by (Addison-Wesley, 2002).